

## 第7章 目标规划习题解答

7.1 解 使用Matlab求解的程序如下

```
clc, clear
a=[1 1]; b=7;
c1=[3 1]'; c2=[1 2]';
lb=zeros(2,1); ub=[5 5]'; %定义线性规划的下界向量和上界向量
[x1,y1]=linprog(-c1,a,b,[],[],lb,ub) %求解第一个线性规划
[x2,y2]=linprog(-c2,a,b,[],[],lb,ub) %求解第二个线性规划
go=-[y1,y2]'; %Matlab工具箱线性规划是求解极小值
goalfun=@(x)[-3*x(1)-x(2);-x(1)-2*x(2)];
[x,fval]=fgoalattain(goalfun,rand(2,1),go,abs(go),a,b,[],[],lb,ub);
x, fval=-fval %恢复到原来的目标函数
```

求得的满意解为  $x_1 = 3.7561$ ,  $x_2 = 3.2439$ , 对应的目标函数的值  $z_1 = 14.5122$ ,

7.2 解 设安排商业节目时间  $x_1$  小时, 新闻节目时间  $x_2$  小时, 音乐节目时间  $x_3$  小时, 该问题的目标规划模型为

$$\begin{aligned} \min z = & p_1(d_1^- + d_1^+ + d_2^- + d_2^+ + d_3^-) + p_2 d_4^- \\ & \begin{cases} x_1 + x_2 + x_3 + d_1^- - d_1^+ = 12, \\ x_1 + d_2^- - d_2^+ = 2.4, \\ x_2 + d_3^- - d_3^+ = 1, \\ 250x_1 - 40x_2 - 17.5x_3 + d_4^- - d_4^+ = 36000, \\ x_1, x_2, x_3, d_1^-, d_1^+, d_2^-, d_2^+, d_3^-, d_3^+, d_4^-, d_4^+ \geq 0. \end{cases} \end{aligned}$$

其中36000为每天收入的上限  $250 \times 60 \times 2.4 = 36000$  美元。

求解的Lingo程序如下

```
model:
sets:
level/1..2/:p,z,goal;
variable/1..3/:x;
s_con_num/1..4/:g,dplus,dminus;
s_con(s_con_num,variable):c;
obj(level,s_con_num)/1 1,1 2,1 3,2 4/:wplus,wminus;
endsets
data:
ctr=?;
goal=? 0;
g=12 2.4 1 36000;
c=1 1 1 1 0 0 0 1 0 250 -40 -17.5;
wplus=1 1 0 0;
wminus=1 1 1 1;
enddata
min=@sum(level:p*z);
p(ctr)=1;
```

```

@for(level(i)|i#ne#ctr:p(i)=0);
@for(level(i):z(i)=@sum(obj(i,j):wplus(i,j)*dplus(j)+wminus(i,j)*dminus(j)));
@for(s_con_num(i):@sum(variable(j):c(i,j)*x(j))+dminus(i)-dplus(i)=g(i));
@for(level(i)|i #lt# @size(level):@bnd(0,z(i),goal(i)));
end

```

求得  $x_1 = 2.4$ ,  $x_2 = 1$ ,  $x_3 = 8.6$ ,  $d_4^- = 35590.5$ ,  $d_4^+ = 0$ , 每天的纯收入为  $36000 - d_4^- = 409.5$  美元。

7.3 解 设在允许的工作时间内产品I生产  $x_1$  件, 产品II生产  $x_2$  件; 在加班时间内产品I生产  $x_3$  件, 产品II生产  $x_4$  件。

建立如下的目标规划模型

$$\begin{aligned}
& \min p_1(d_1^- + d_2^-) + p_2 d_3^- \\
& \begin{cases} 3x_1 + 2.5x_2 + d_1^- = 120, \\ 3x_1 + 2.5x_2 + 3x_3 + 2.5x_4 + d_2^- = 160, \\ 10x_1 + 8x_2 + 8.5x_3 + 7x_4 + d_3^- = 640, \\ d_i^- \geq 0, i = 1, 2, 3; x_i \geq 0 \text{ 且为整数}, i = 1, 2, 3, 4. \end{cases}
\end{aligned}$$

其中第3个约束右边的640为利润的上界, 由于无论生产产品I或II, 每小时的赢利不超过4元, 每周的生产时间不超过160小时, 因而最大利润不超过640元。

计算的Lingo程序如下

```

model:
sets:
level/1..2/:p,z,goal;
variable/1..4/:x;
s_con_num/1..3/:g,dminus;
s_con(s_con_num,variable):c;
obj(level,s_con_num)/1 1,1 2,2 3/:wminus;
endsets
data:
ctr=?;
goal=? 0;
g=120 160 640;
c=3 2.5 0 0 3 2.5 3 2.5 10 8 8.5 7;
wminus=1 1 1;
enddata
min=@sum(level:p*z);
p(ctr)=1;
@for(level(i)|i#ne#ctr:p(i)=0);
@for(level(i):z(i)=@sum(obj(i,j):wminus(i,j)*dminus(j)));
@for(s_con_num(i):@sum(variable(j):c(i,j)*x(j))+dminus(i)=g(i));
@for(level(i)|i #lt# @size(level):@bnd(0,z(i),goal(i)));
@for(variable:@gin(x));
end

```

求得  $x_1 = 40$ ,  $x_2 = 0$ ,  $x_3 = 10$ ,  $x_4 = 4$ ,  $d_1^- = 0$ ,  $d_2^- = 0$ ,  $d_3^- = 127$ , 即产品I生产50件, 产品II生产4件时, 总的利润最大, 最大利润为  $640 - d_3^- = 413$  元。